

PHREGプロシジャにおける

共変量調整解析に関連したオプション機能

○魚住 龍史¹ 矢田 真城² 浜田 知久馬²

¹京都大学大学院医学研究科 医学統計生物情報学

²東京理科大学大学院 工学研究科 経営工学専攻

Investigating fascinating aspects associated with covariate-adjusted analysis using PHREG procedure

Ryuji Uozumi¹, Shinjo Yada², and Chikuma Hamada²

¹*Department of Biomedical Statistics and Bioinformatics, Kyoto University Graduate School of Medicine*

²*Department of Management Science, Graduate School of Engineering, Tokyo University of Science*

要旨

PHREG プロシジャは Cox の比例ハザードモデルによる解析を実施するためのプロシジャであり, SAS/STAT のバージョンアップのたびに, 構文数およびオプション数が増加している. 本稿では, 『生存時間解析 応用編 –SAS による生物統計』で取りあげられている PHREG プロシジャにおけるオプション機能のうち, BASELINE 文および LSMEANS 文に着目した. これらの有用なオプション機能として, BASELINE 文における DIRADJ オプションの機能および LSMEANS 文による Diffogram の作成に対する理解をより深めるための報告を行う.

キーワード: 生存時間解析 比例ハザードモデル PHREG BASELINE 生存関数 Breslow 法 調整生存関数 DIRADJ LSMEANS Diffogram Mean-Mean scatter plot ODS GRAPHICS

1 はじめに

大橋・浜田 (1995) が刊行されてから 20 年もの歳月が流れ, 2016 年 7 月に姉妹本という位置付けで, 『生存時間解析 応用編 –SAS による生物統計』が刊行された (大橋ら, 2016). 現在, SAS 9.4 が最新版としてリリースされているが, LIFETEST プロシジャなどの生存時間解析に関連したプロシジャは SAS/STAT として提供されており, ほぼ毎年メンテナンス版がリリースされている. 『生存時間解析 応用編 –SAS による生物統計』では, 主に SAS/STAT 9.2 以降に追加された最新機能の使い方を幅広く解説しており, 2016 年の SAS ユーザー総会の前夜祭の日に, SAS Learning Session として解説も行われた (浜田・魚住, 2016a). また, 『生存時間解析 応用編 –SAS による生物統計』で取りあげられていない生存時間解析に関連した最新機能も SAS

ユーザー総会で多く報告されており (浜田・魚住, 2016b; 魚住ら, 2016b; 矢田・浜田, 2016; 矢田ら, 2016, 2017), 今後も新たなプロシジャおよびオプション機能が追加されることが期待されている。

本稿では、『生存時間解析 応用編 -SASによる生物統計』の刊行後に、読者から寄せられた質問のうち、PHREG プロシジャに関連した内容に焦点を当てることとする。PHREG プロシジャはCoxの比例ハザードモデル (Cox, 1972) による解析を実施するためのプロシジャであり、SAS/STATのバージョンアップのたびに、ステートメントおよびオプションの数が増加している。大橋・浜田 (1995) 刊行時は10のステートメント (文) が提供されているだけであったが、SAS/STAT 14.2においては24のステートメントが提供されている (SAS Institute Inc., 2016)。著者らは、PHREG プロシジャにおけるオプション機能のうち、ベースライン生存関数を推定するためのBASELINE文、および固定効果に対する最小二乗平均を求めるLSMEANS文におけるオプション機能に着目した。BASELINE文では、SAS/STAT 9.2よりDIRADJオプションが追加されており (SAS Institute Inc., 2009)、平均生存関数の推定が可能である。また、ODS GRAPHICS ONの状態では、LSMEANS文を指定してPHREGプロシジャを実行すると、水準間の対比較の結果を表すDiffogramが作成される。本稿では、BASELINE文およびLSMEANS文における指定方法および有用なオプション機能の解説を行い、指定方法によって異なる出力結果が得られる点に関して、理解をより深めるための報告を行う。

2 BASELINE文とオプション機能

PHREGプロシジャにおいて、BASELINE文は大橋・浜田 (1995) が刊行された頃からサポートされている。BASELINE文を指定することで、ベースラインハザード関数やベースライン生存関数などの推定を行うことが可能である。

本節では、BASELINE文を用いてベースライン生存関数の推定を行うことを考える。対象データとして、肺癌のデータ (データセット名: VALung) を用いる。データセット VALung (Veterans Administration Lung cancer trial) は、Kalbfleisch and Prentice (2002) で使用されたデータを一部抽出したものであり ($n = 137$)、SAS/STAT PHREGプロシジャのマニュアル、および大橋ら (2016) 第3章においても用いられている。この研究の目的は、男性の進行性肺癌患者を対象としたランダム化比較試験であり、治療法 (変数名: Therapy) として、試験治療 (Therapy = 'test') と標準治療 (Therapy = 'standard') を比較するために行われた。評価項目は死亡までの時間 (日) (変数名: Time) で、共変量の1つとして組織型 (変数名: Cell) が挙げられる。組織型は4水準 (Cell = 'adeno' 「腺癌」, 'small' 「小細胞癌」, 'large' 「大細胞癌」, 'squamous' 「扁平上皮癌」) のカテゴリカル変数である。なお、PHREGプロシジャの処理の都合上、文字変数であるTherapyおよびCellの代わりに、治療法を表す数値変数Therapy2 (Therapy2 = 1「試験治療」, 2「標準治療」), 組織型を表す数値変数CellC (CellC = 1「腺癌」, 2「大細胞癌」, 3「小細胞癌」, 4「扁平上皮癌」) も用意した。

2.1 BASELINE文による生存関数の推定

データセット VALung に対して、LIFETESTプロシジャおよびPHREGプロシジャで生存時間曲線を描くと、SGPANELプロシジャでパネルを代えて出力すると図1のように示される。

なお、本稿において示すプロットは、モノクロ印刷されても識別しやすいよう、ODSの出力としてジャーナルスタイル "STYLE = JOURNAL" を指定している。例えば、群別のグラフを示す場合、デフォルトでは群の違いが色で分けて出力されてしまう。ジャーナルスタイルの場合、モノクロ印刷で群の違いを識別できるよう、実線と破線で分けて出力される。

また、図1のように、SAS/STATのプロシジャの実行結果をデータセットとして作成し、SG (Statistical Graphics) プロシジャで出力する方法については、これまでのSASユーザー総会でも多くの報告が行われている(魚住・浜田, 2011; 魚住・浜田, 2012; 魚住ら, 2016a)。SGPANEL プロシジャによる図1の詳細なプログラムは付録Aを参照されたい。

SAS プログラム (詳細は付録Aを参照)	<pre>ods graphics on; ods output SurvivalPlot=lifetest00; proc lifetest data=VALung plots=s method=breslow; time Time*Censor(1); strata Cell; run; data DummyCell; do CellC=1 to 4; output; end; run; ods output SurvivalPlot=phreg00; proc phreg data=VALung plots(overlay)=survival; class CellC / param=glm; model Time*Censor(1) = CellC; baseline out=phregout00 survival=S covariates=DummyCell / nomean group=CellC; run;</pre>
出力結果	

図1: LIFETEST プロシジャおよび PHREG プロシジャによる作成プログラムと組織型別の生存時間曲線

図1における PHREG プロシジャのプログラムでは、BASELINE 文における OUT = オプション、あるいは ODS GRAPHICS の機能を用いた ODS OUTPUT SURVIVALPLOT = によって出力される。いずれのデータセットを用いても、同様の生存関数の推定結果が得られている。

図1の生存時間曲線は、LIFETEST および PHREG のいずれのプロシジャから作成しても、Breslow 法による生存関数の推定を行っている。

ここで、 (T, D, X) を生存時間、イベントの有無、説明変数を表す確率変数として、 (t_i, d_i, x_i) はこれらの実現値を表し、 $i = 1, \dots, n$ は互いに独立であることを仮定する。ただし、 $d_i = 1$ はイベント、 $d_i = 0$ は打ち切りを表す指示関数とする。さらに、 $Y_i(t) = 1(t_i \geq t)$ を個体 i が時点 t においてリスク集合に含まれることを表す指示関数、 $N_i(t) = 1(t_i \leq t, d_i = 1)$ を個体 i の時点 t におけるイベントを表す指示関数とし、 $\bar{Y}(t) = \sum_i Y_i(t)$ を時点 t におけるリスク集合の大きさを表す指示関数、 $\bar{N}(t) = \sum_i N_i(t)$ を時点 t までの累積イベント数、 $\Delta\bar{N}(t)$ を時点 t におけるイベント数とする。

LIFETEST プロシジャでは、デフォルトの Kaplan-Meier 法による生存関数は

$$\hat{S}(t) = \prod_{t_i \leq t} \left(1 - \frac{\Delta\bar{N}(t_i)}{\bar{Y}(t_i)} \right)$$

と推定される (Kaplan and Meier, 1958)。一方、Breslow 法による生存関数は、Nelson-Aalen 法による累積ハザード関数 (Nelson, 1969; Aalen, 1978) を用いて、

$$\hat{S}(t) = \exp\left(-\int_0^t \frac{d\bar{N}(t_i)}{\bar{Y}(t_i)}\right) \approx \exp\left(-\sum_{t_i \leq t} \frac{\Delta\bar{N}(t_i)}{\bar{Y}(t_i)}\right)$$

と推定される。

一方、PHREG プロシジャを用いて Breslow 法による推定を考える。説明変数ベクトル X に対する比例ハザードモデル

$$\lambda(t | X) = \lambda_0(t) \exp(\beta^T X)$$

を考えると、ベースラインハザード関数が

$$\hat{\lambda}_0(t) = \frac{\Delta\bar{N}(t)}{\sum_i Y_i(t) \exp(\beta^T x_i)}$$

と推定される (Breslow, 1974)。このとき、累積ベースラインハザード関数は

$$\hat{\Lambda}_0(t) = \sum_{t_i \leq t} \hat{\lambda}_0(t_i)$$

となり、Breslow 法による生存関数は

$$\hat{S}(t | x, \hat{\beta}) = \exp(-\hat{\Lambda}_0(t) \exp(\hat{\beta}^T x))$$

と推定される。また、Kaplan-Meier 法による生存関数も

$$\hat{S}(t | x, \hat{\beta}) = \prod_{t_i \leq t} \left(1 - \exp(\hat{\beta}^T x) \hat{\lambda}_0(t_i) \right)$$

と同じように推定できる。

2.2 DIRADJ オプションを用いた生存関数の推定

図1の PHREG プロシジャでは、単変量の比例ハザードモデルに対して、BASELINE 文を用いて生存時間曲線を推定した。PROC PHREG 文において、PLOTS(OVERLAY) = SURVIVAL と指定しているものの、実際は4群が1つのプロットとして描かれない。しかし、BASELINE 文において DIRADJ オプションを指定すると、図2のように PHREG プロシジャの ODS GRAPHICS による機能として、4群を1つのプロットに集約し

た生存時間曲線が出力される。なお、図2においては、単変量の比例ハザードモデルであることから、BASELINE 文において COVARIATES = を指定しなくても、図1と同様の出力が得られる。

BASELINE 文で DIRADJ オプションを指定して、図2のように4群を1つのプロットに集約した生存時間曲線が出力したが、DIRADJ では COVARIATES = で指定した変数で調整した生存関数を推定している。もし COVARIATES = を指定しないで実行した場合、GROUP = で指定した変数の各水準に対する調整生存関数として、GROUP = で指定していない変数の情報に基づいた重み付き生存関数が推定される。




図2: DIRADJ オプションを用いた PHREG プロシジャによるプログラムと ODS GRAPHICS による出力

なお、もし図2のプログラムにおいて、BASELINE 文のオプションで METHOD = KM を指定しても、DIRADJ オプションは Breslow 法による生存関数の推定のみ対応しており、自動的に METHOD = BRESLOW と実行される。DIRADJ オプションは SAS/STAT 12.1 以降でサポートされている機能である (SAS Institute Inc., 2012)。

ここで、わかりやすく比例ハザードモデルの説明変数として、組織型に加えて治療を表す2水準の変数も加えた場合で考え、各治療に対する調整生存関数を推定したいとする。このとき、DIRADJ および COVARIATES = をともに指定した場合は図3のプログラムとなる。

```

data Dummy; do CellC=1 to 4; do Therapy2=1 to 2; output; end; end; run;
proc phreg data=VALung plots(overlay)=survival;
  class CellC Therapy2 / param=glm;
  model Time*Censor(1) = CellC Therapy2;
  baseline out=out01 survival=S covariates=Dummy / diradj nomean group=Therapy2;
run;

```

図 3: DIRADJ および COVARIATES = オプションを用いた調整生存関数の推定プログラム

各治療 i において組織型 j 別に推定される (時点 t における) 生存関数を \hat{S}_{ij} , 例数を n_{ij} とすると, 表 1 のように表記できる.

表 1: 各水準における推定生存関数

	Cell (j)			
Treatment (i)	\hat{S}_{11}	\hat{S}_{12}	\hat{S}_{13}	\hat{S}_{14}
	\hat{S}_{21}	\hat{S}_{22}	\hat{S}_{23}	\hat{S}_{24}

図 3 のプログラムでは, 時点 t における治療 i に対する調整生存関数は

$$\bar{S}_i = \sum_j \hat{S}_{ij} / J$$

と推定されていることになる. ただし, J は調整する組織型の水準数であり, 今回のデータでは $J=4$ である.

```

proc phreg data=VALung plots(overlay)=survival;
  class CellC Therapy2 / param=glm;
  model Time*Censor(1) = CellC Therapy2;
  baseline out=out02 survival=S / diradj nomean group=Therapy2;
run;

```

図 4: DIRADJ オプションを用いた重み付き調整生存関数の推定プログラム

一方, 図 4 のプログラムでは, 時点 t における治療 i に対する重み付き調整生存割合は

$$\tilde{S}_i = \sum_j n_{.j} \hat{S}_{ij} / n_{..}$$

と推定される. ただし, $n_{..}$ はデータセット VALung の総例数, $n_{.j}$ は組織型 j における総例数である. 同じ PHREG プロシジャ内の機能で例えるならば, LSMEANS 文における OM オプションに該当する内容といえる. データセット VALung の場合, 組織型を表す変数の各水準における例数は adeno: $n_1 = 27$, large: $n_2 = 27$, small: $n_3 = 48$, squamous: $n_4 = 35$ となるため, 時点 t における治療 i に対する重み付き調整生存割合は

$$\tilde{S}_i = \frac{27}{137} \hat{S}_{i1} + \frac{27}{137} \hat{S}_{i2} + \frac{48}{137} \hat{S}_{i3} + \frac{35}{137} \hat{S}_{i4}$$

となる.

3 LSMEANS 文とオプション機能

LSMEANS 文は GLM プロシジャや MIXED プロシジャにおいてサポートされていた機能で、PHREG プロシジャにおいては SAS/STAT 9.3 から新たに追加された (SAS Institute Inc., 2011). GLM プロシジャからの類推で LSMEANS 「最小二乗平均」という言葉が使われているが、推定は最尤法で行われている (大橋ら, 2016). なお、LSMEANS 文ではすべての水準間の最小二乗平均の対比較が行われるため、対比較の数が多い場合は LSMESTIMATE 文や ESTIMATE 文で特定の線型仮説を明示的に指定することが有用である (浜田, 2013; 魚住, 2014).

3.1 LSMEANS 文による Diffogram の作成

ODS GRAPHICS ON の状態で LSMEANS 文を指定して PHREG プロシジャを実行すると、「最小二乗平均」の水準間の比較結果が出力される. 『生存時間解析 応用編 -SAS による生物統計』の 3.2 節におけるプログラム 3.2.1 および出力結果を図 5 に示している. 比例ハザードモデルの説明変数として、Kps, Cell, Prior, Therapy, Therapy と Prior の交互作用を含めている. なお、Kps はカルノフスキー (Karnofsky) の Performance Scale を表す連続変数、Prior は既往歴を表す 2 値カテゴリカル変数である.




図 5: LSMEANS 文を用いた Diffogram 作成プログラムとその出力

LSMEANS 文を指定して PHREG プロシジャを実行すると、水準間の対比較の結果が得られる。図 5 は ODS SELECT DIFFPLOT により出力の指定が行われており、Mean-Mean scatter plot あるいは Diffogram と呼ばれる (High, 2014)。LSMEANS 文で 4 水準の変数 Cell を指定したため、 ${}_4C_2 = 6$ 通りの比較結果として、各群の最小二乗平均の 95%信頼区間として 6 本描かれている。対角線が帰無仮説を表しており、信頼区間と交わっているかで有意かどうか判定している。有意な対比較の信頼区間は実線、有意でない対比較の信頼区間は破線で示されている。

図 5 の Diffogram より、adeno-squamous 「腺癌と扁平上皮癌」、small-squamous 「小細胞癌と扁平上皮癌」の比較においては、Tukey-Kramer 法による多重性の調整を行って有意な結果が得られている (多重比較については、浜田 (2015) を参照されたい)。なお、この結果は比例ハザードモデルで組織型以外の変数で調整を行っているため、共変量および多重性の二重の調整を行っていることになる。

ここで、表 2 は図 5 のプログラムを実行して得られる出力結果で、ODS SELECT DIFFS の指定により出力された表である。しかし、「最小二乗平均の差」と出力されており、図 5 の Diffogram がどのように作成されたか疑問に思うのではないだろうか。

表 2 : LSMEANS 文による共変量を調整した推定結果

		Cell の最小 2 乗平均の差 多重比較の調整 : Tukey-Kramer					
Cell	_Cell	推定値	調整済		指数	調整済	
			下限値	上限値		下限 Exp	上限 Exp
adeno	large	0.7491	-0.03358	1.5317	2.1150	0.9670	4.6261
adeno	small	0.3064	-0.3906	1.0034	1.3586	0.6766	2.7277
adeno	squamous	1.1605	0.3898	1.9313	3.1916	1.4766	6.8983
large	small	-0.4427	-1.1149	0.2296	0.6423	0.3279	1.2581
large	squamous	0.4114	-0.3158	1.1387	1.5090	0.7292	3.1227
small	squamous	0.8541	0.1632	1.545	2.3493	1.1773	4.6880

3.2 LSMEANS 文におけるオプション機能を用いた Diffogram の作成

図 6 は、LSMEANS 文において、図 5 よりも多くのオプションを指定したプログラム、およびその出力結果を示している。

まず、LSMEANS 文において MEANS オプションを指定することで、「最小二乗平均の差」の出力に加えて、表 3 の最小二乗平均の推定値を出力することができる。PHREG プロシジャの場合、対数ハザードが推定されていることを意味しており、「最小二乗平均の差」は対数ハザード比を推定していることに相当する。

次に、LSMEANS 文における PLOT = で ODS GRAPHICS の機能で出力させるグラフを明示的に指定している。PLOT = DIFF を指定しているので、図 5 に示したような Diffogram が作成される。PLOT = DIFF はデフォルトの指定である。さらに、PLOT = DIFF のオプションとして、NOABS および CENTER を指定している。図 5 の出力では、デフォルトである ABS オプションの機能により、最小二乗平均の推定値がすべて左上にプロットされるよう、縦軸・横軸のいずれを対照水準とするか対比較ごとに変動させて出力させていた。これ

に対して、NOABS を指定することで、縦軸・横軸のいずれを対照水準とするか固定して出力している。また、CENTER オプションを指定することで、最小二乗平均の推定値を○で特定して示している。

以上のように、MEANS オプションを指定することで、表 3 の最小二乗平均の推定値が出力されるため、図 5 および図 6 の出力で示した Diffogram がどのように作成されたか確認できるであろう。

なお、MEANS オプションも指定した下で、PLOT = DIFF の代わりに PLOT = MEANPLOT を指定すると、LSMEANS 文で指定した変数に対する各水準の最小二乗平均の推定値とその 95%信頼区間のグラフを作成できる (付録 B を参照されたい)。




図 6 : LSMEANS 文のオプション機能を用いた Diffogram 作成プログラムとその出力

表 3 : LSMEANS 文による共変量を調整した推定結果

Cell	推定値	Cell の最小 2 乗平均			Exponentiated Lower	Exponentiated Upper
		下限	上限	指数		
adeno	-0.5778	-1.5253	0.3697	0.5611	0.2176	1.4472
large	-1.3269	-2.2197	-0.4341	0.2653	0.1086	0.6479
small	-0.8843	-1.8132	0.04472	0.413	0.1631	1.0457
squamous	-1.7384	-2.4919	-0.9848	0.1758	0.08275	0.3735

3 まとめ

本稿では、比例ハザードモデルによる生存時間解析を行うためのプロシジャである PHREG プロシジャに関して、『生存時間解析 応用編 –SAS による生物統計』の刊行後に読者から寄せられた質問事項として、ベースライン生存関数を推定するための BASELINE 文、および固定効果に対する最小二乗平均を求める LSMEANS 文におけるオプション機能に着目した。BASELINE 文では、SAS/STAT 9.2 より DIRADJ オプションが追加されており、COVARIATE = オプションと併用することで調整生存関数の推定が可能である。COVARIATE = オプションを指定せずに DIRADJ オプションを指定すれば、調整する変数の各水準における例数で重み付けした調整生存関数の推定が行われていることに相当する。なお、ODS GRAPHICS ON の状態で、PROC PHREG 文において PLOTS(OVERLAY) = SURVIVAL および BASELINE 文で DIRADJ オプションを指定すれば、すべての群の生存関数の推定結果のグラフとして出力される。このような調整生存関数の推定は観察研究の論文報告でよく用いられており、例えば Yuan et al. (2017) において図として報告されている生存時間曲線が該当する。一方、ODS GRAPHICS ON の状態で、LSMEANS 文を指定して PHREG プロシジャを実行すると、水準間の対比較の結果を表す Diffogram が作成される。Diffogram の縦軸・横軸の値には、最小二乗平均の推定値 (PHREG プロシジャの場合は対数ハザードに相当する) が出力されており、LSMEANS 文において MEANS オプションを指定することで確認可能である。その他にも、Diffogram の解釈を役立てる機能として、LSMEANS 文における NOABS および CENTER の有用性を示した。

本稿で取りあげたすべての出力結果は、付録 A を含めた本稿に記述されている SAS プログラムで再現可能である。生存時間解析に従事する国内外の多くの実務家にとって、本稿の報告内容がお役に立てれば幸いである。

参考文献

- [1] Aalen O. Nonparametric estimation of partial transition probabilities in multiple decrement models. *Annals of Statistics*. **6**:701–726, 1978.
- [2] Breslow N. Covariance analysis of censored survival data. *Biometrics*. **30**:89–99, 1974.
- [3] Cox DR. Regression models and life-tables (with Discussion). *Journal of the Royal Statistical Society, Series B*. **34**:187–220, 1972.
- [4] High R. Plotting Differences among LSMEANS in Generalized Linear Models. *Proceedings of the SAS Global Forum*. SAS Institute Inc., Cary, NC, 2014. Available at <http://support.sas.com/resources/papers/proceedings14/1902-2014.pdf>.
- [5] Kaplan EL, Meier P. Nonparametric estimation from incomplete observations. *Journal of the American Statistical Association*. **53**:457–481, 1958.
- [6] Klein JP, van Houwelingen HC, Ibrahim JG, Scheike TH. *Handbook of Survival Analysis*. Chapman and Hall/CRC, Boca Raton, FL, 2013.
- [7] Nelson W. Hazard plotting for incomplete failure data. *Journal of Quality Technology*. **1**:27–52, 1969.
- [8] SAS Institute Inc. *SAS/STAT(R) 9.2 User's Guide* (2nd edn.). SAS Institute Inc., Cary, NC, 2009.
- [9] SAS Institute Inc. *SAS/STAT(R) 9.3 User's Guide*. SAS Institute Inc., Cary, NC, 2011.

- [10] SAS Institute Inc. *SAS/STAT(R) 12.1 User's Guide*. SAS Institute Inc., Cary, NC, 2012.
- [11] SAS Institute Inc. *SAS/STAT(R) 14.2 User's Guide*. SAS Institute Inc., Cary, NC, 2016.
- [12] Yuan C, Morales-Oyarvide V, Babic A, Clish CB, Kraft P, Bao Y, Qian ZR, Rubinson DA, Ng K, Giovannucci EL, Ogino S, Stampfer MJ, Gaziano JM, Sesso HD, Cochrane BB, Manson JE, Fuchs CS, Wolpin BM. Cigarette smoking and pancreatic cancer survival. *Journal of Clinical Oncology*. **35**:1822–1828, 2017.
- [13] 魚住龍史・浜田知久馬. SG (Statistical Graphics) Procedures による Kaplan-Meier プロットの作成. SAS ユーザー総会 論文集 2011, 185–199.
- [14] 魚住龍史・浜田知久馬. がん臨床試験における腫瘍縮小効果の検討に有用なグラフの作成 –SGPLOT プロシージャの最新機能を活用–. SAS ユーザー総会 論文集 2012, 151–165.
- [15] 魚住龍史. LS-Means 再考 –GLM と PLM によるモデル推定後のプロセス–. SAS ユーザー総会 論文集 2014, 449–463.
- [16] 魚住龍史・森田智視. 生存時間解析における三種の神器. 呼吸, **34**: 1083–1089, 2015.
- [17] 魚住龍史・吉田早織・平井隆幸, 浜田知久馬. Kaplan-Meier プロット・Forest プロット作成の応用: グラフ出力範囲内・範囲外への数値出力. SAS ユーザー総会 論文集 2016a, 216–232.
- [18] 魚住龍史・矢田真城・浜田知久馬. SAS プロシージャを用いた生存時間データに対する例数設計の変革. SAS ユーザー総会 論文集 2016b, 250–267.
- [19] 魚住龍史. 有名論文から統計の基礎を学ぶ: COURAGE 試験における生存時間解析の側面. *Coronary Intervention*, **13**: 38–43, 2017.
- [20] 大橋靖雄・浜田知久馬. 生存時間解析 –SAS による生物統計. 東京大学出版会, 1995.
- [21] 大橋靖雄・浜田知久馬・魚住龍史. 生存時間解析 応用編 –SAS による生物統計. 東京大学出版会, 2016.
- [22] 浜田知久馬. SAS 生存時間解析プロシージャの最新の機能拡張. SAS ユーザー総会 論文集 2013, 3–72.
- [23] 浜田知久馬. SAS による多重比較「美女と野獣」の統計学. SAS ユーザー総会 論文集 2015, 367–388.
- [24] 浜田知久馬・魚住龍史. SAS による生存時間解析の拡張機能. SAS Learning Session 2016a. 配付資料.
- [25] 浜田知久馬・魚住龍史. SAS による生存時間分布の予測「Death Note の統計学」. SAS ユーザー総会論文集 2016b, 297.
- [26] 矢田真城・浜田知久馬. ガンマ過程を用いた比例ハザードモデルによる解析. SAS ユーザー総会 論文集 2016, 100–109.
- [27] 矢田真城・魚住龍史・浜田知久馬. 生存時間データに対するベイズ流例数設計. SAS ユーザー総会 論文集 2016, 268–278.
- [28] 矢田真城・魚住龍史・浜田知久馬. 区分指数モデル及びフレイルティモデルに対するベイズ流生存時間解析. SAS ユーザー総会 論文集 2017.

連絡先

E-mail : uozumi@kuhp.kyoto-u.ac.jp

付録 A: 図 1 の SGPPANEL プロシジャによる出力のためのプログラム詳細

```

*=====  

* フォーマット ;  

proc format;  

  value Cellf 1='adeno' 2='large' 3='small' 4='squamous'; value procf 1='LIFETEST' 2='PHREG';  

run;  

*=====  

* LIFETESTプロシジャによる出力結果 ;  

*=====  

ods graphics on;  

ods listing close;  

ods output SurvivalPlot=lifetest00;  

proc lifetest data=VALung plots=s method=breslow;  

  time Time*Censor(1); strata Cell;  

run;  

*=====  

* PHREGプロシジャによる出力結果;  

*=====  

data DummyCell; do CellC=1 to 4; output; end; run;  

ods output SurvivalPlot=phreg00;  

proc phreg data=VALung plots(overlay)=survival;  

  class CellC / param=glm; model Time*Censor(1) = CellC;  

  baseline out=phregout00 survival=S covariates=DummyCell / nomean group=CellC;  

run;  

ods listing;  

ods graphics off;  

*=====  

* SGPPANELプロシジャによる図1の出力 ;  

*=====  

data lifetest00;set lifetest00; proc=1; run;  

data phreg00;set phreg00; proc=2; rename Vector_ =StratumNum; run;  

data survival00; set lifetest00 phreg00; run;  

ods html image dpi=400 style=journal;  

ods graphics / reset noborder noscale width=600px height=400px imagefmt=png; ods listing close;  

proc sgpanel data=survival00 noautolegend;  

  panelby proc / novarname;  

  step x=Time y=Survival / group=StratumNum name='Survival';  

  scatter x=Time y=Censored / group=StratumNum markerattrs=(symbol=plus);  

  rowaxis values=(0 to 1 by 0.1) label='Survival'; colaxis values=(0 to 1000 by 200) label='Time';  


  keylegend 'Survival' / noborder position=bottom; format StratumNum Cellf. proc procf.;  

run;  

ods listing; ods graphics off;  

ods html close;

```



付録 B: LSMEANS 文のオプション機能を用いた最小二乗平均の推定結果のグラフ出力